Departament de Cristal.lografia,
Mineralogia i Dipòsits Minerals.

GEOLOGIA I METAL.LOGENIA DEL CONTACTE SUD DEL GRANIT D'ANDORRA (PIRINEU CENTRAL).

Albert SOLER i GIL

Gener 1990

ANNEX 6

EL PROGRAMA CLORITAP

Programa cloritap.bas. Autor: Ayora, C. i Soler, A.

Aquest programa calcula la fO2 i FS2 a partir de les dades termodinamiques dels diferents components de les clorites. Basat en Walshe (1986).

```
10 REM CLORITAP= PROGRAMA DEFINITIU FUGACITAT D'O2 A PARTIR DE FE2/FE3
20 DIM A(10),B(10),C(10),D(10),E(10),M(10),N(10),P(10),Q(10)
30 REM
40 CLS: LOCATE 10,10: INPUT "MOSTRA= ",M$
45 IF M$="" THEN 2150
50 LPRINT CHR$(27): LPRINT "MOSTRA= ";M$
60 LPRINT CHR$(27)
70 REM ENTRADA DE DADES D'ANALISI CLORITA
80 FOR I=1 TO 9: A(I)=0: NEXT I
90 LOCATE 14,10:INPUT "SiO2= ",A(1)
100 LOCATE 15,10: INPUT "Al2O3= ",A(2)
110 LOCATE 16,10: INPUT "MgO= ",A(3)
120 LOCATE 17,10: INPUT "FeO= ",A(4)
130 LOCATE 18,10: INPUT "MnO= ",A(6)
140 LOCATE 20,10: INPUT "Fe3+/FeT= ",RF: RM=RF
145 LOCATE 21,10: INPUT "TEMPERATURA (C)= ",TM: TM=TM+273.15
150 FOR I=1 TO 8: A(9)= A(9)+A(I): NEXT I
160 REM
170 REM PROPORCIONS MOLARS
180 B(1)= A(1)/60.086
190 B(2)= A(2)/101.963
200 B(3)= A(3)/40.312
210 B(4)= A(4)/71.847
220 B(6)= A(6)/70.93801
230 B(8)= A(8)/18
240 REM
250 REM NOMBRE D'OXIGENS
260 C(1)= B(1)*2
270 C(2)= B(2)*3
280 FOR I=3 TO 8: C(I)= B(I): NEXT I
290 C(9)=0
300 FOR I= 1 TO 8: C(9)= C(9)+C(I): NEXT I
310 REM
320 REM SUBINDEX FORMULA ESTRUCTURAL
330 FOR I=1 TO 8: D(I)= B(I)*14/C(9): NEXT I
340 D(2)= D(2)*2
350 D(8)= D(8)*2
360 D(3)= D(3)+D(6)
370 D(9)=0
380 FOR I=1 TO 4:  D(9)=D(9)+D(I): NEXT I
390 REM
391 E(6)= D(4)*RM
393 F= 28/(28+E(6))
410 E(1)= D(1)*F
420 E(2)= 4-F*D(1)
430 E(3)= F*D(3)
440 E(6)= F*E(6)
450 E(4)= F*D(4)*(1-RM)
460 E(7)= F*(D(1)+D(2))-4
470 REM
480 REM CALCUL DE L'EQUACIO (43) DE WALSHE (1986)
490 IF D(9)<=10 THEN 510
```

```
500 D(9)=10
510 LA5= LOG(5-F*D(9)/2)
520 LA1= 6*LOG(E(3)/6) +2*LOG((E(1)-2)/2)
530 LA2= LOG(59.72) +5*LOG(E(3)/6) +LOG(E(7)/6) +LOG((E(1)-2)/2) +LOG(E(2)/2)
570 REM
580 REM
590 REM CALCUL DE L'EQUACIO (44) DE WALSHE (1986)
610 LA3= LOG(59.72) +5*LOG(E(4)/6) +LOG(E(7)/6) +LOG((E(1)-2)/2) +LOG(E(2)/2)
620 LA4= 0
680 REM CALCUL DE PROPORCIONS MOLARS DE COMPONENTS
690 X6=E(6)
700 X3=(E(4)-4*X6)/5
710 X4=0
720 X5= 5-F*D(9)/2
721 X2= E(7)-(X3+4*X5+X6)
723 IF X2>=(1-(X6+X3+X5)) THEN 727
724 X1=(E(3)-5*X2)/6
725 GOTO 740
727 X2=1-(X6+X3+X5): X1=0
730 X1= (E(3)-5*X2)/6
740 CLS
1240 REM
1241 ALFA= - 13530*EXP(-5026/TM)
1242 BETA= 35600!*EXP(-4845/TM)
1243 GAMA= ALFA*(1-X6) +BETA*(1-X6)*(1-X6)/2
1244 GAMA= GAMA*LOG(10)
1245 REM ARA HE SUPOSAT QUE GAMA ES ZERO, ES A DIR A(6)=X6
1246 LA6= LOG (X6)
1250 REM CALCUL FO2
1260 LAO2= LA6/LOG(10)-LA3/LOG(10)+2.045-6905/TM
1270 LAO2=LAO2*4
1280 REM CALCUL AS2 EN EQUILIBRI AMB PY
1290 LAS2= 5*LAO2 + LA5/LOG(10)-2*LA3/LOG(10)+ 63.107 - 4095.6/TM
1300 LAS2= LAS2/10
1310 LAS6= 11*LAO2/2 + LA5/LOG(10)-2*LA6/LOG(10)+ 59.063 + 9688.2/TM
1320 LAS6= LAS6/10
1330 REM CALCUL LOG AS2 EQUILIBRI AMB PO
1340 LASP3= LA5/LOG(10)+5*LAO2-2*LA3/LOG(10)+69149.4/TM-9.939999
1350 LASP3= LASP3/5
1360 LASP6= LA5/LOG(10)+11*LAO2/2-2*LA6/LOG(10)+82933.21/TM-13.99
1370 LASP6= LASP6/5
1380 REM
1390 REM
1400 REM ESCRIURE RESULTATS
1410 REM
1420 LPRINT "CLORITA":LPRINT
1430 LPRINT "SiO2=  ";A(1)
1440 LPRINT "Al2O3= ";A(2)
1450 LPRINT "MgO=   ";A(3)
1460 LPRINT "FeO=   ";A(4)
1470 LPRINT "MnO=   ";A(6)
1480 LPRINT
1490 LPRINT "Si(T)=    ";E(1)
1500 LPRINT "Al(T)=    ";E(2)
1510 LPRINT "Fe3+(T)=  ";E(5)
1520 LPRINT "TOTAL(T)= ";E(1)+E(2)+E(5)
1530 LPRINT "Al(O)=    ";E(7)
1540 LPRINT "Mg(O)=    ";E(3)
```

```
1550 LPRINT "Fe3+(O)=  ";E(6)
1560 LPRINT "Fe2+(O)=  ";E(4)
1570 LPRINT "TOTAL(O)= ";E(7)+E(3)+E(6)+E(4)
1580 LPRINT
1590 LPRINT "FRACCIONS MOLARS"
1600 LPRINT "X1= ";X1
1610 LPRINT "X2= ";X2
1620 LPRINT "X3= ";X3
1630 LPRINT "X4= ";X4
1640 LPRINT "X5= ";X5
1650 LPRINT "X6= ";X6
1660 LPRINT "TOTAL= ";X1+X2+X3+X4+X5+X6
1670 LPRINT
1680 LPRINT "Mg/(Mg+Fe2+Fe3)= ";E(3)/(E(3)+E(5)+E(4)+E(6))
1690 LPRINT "Fe3/(Fe3+Fe2)=   ";(E(5)+E(6))/(E(5)+E(6)+E(4))
1700 LPRINT
1710 LPRINT "ACTIVITATS DELS COMPONENTS"
1720 LPRINT "Log A1= ";LA1/LOG(10)
1730 LPRINT "Log A2= ";LA2/LOG(10)
1740 LPRINT "Log A3= ";LA3/LOG(10)
1750 LPRINT "Log A4= ";LA4/LOG(10)
1760 LPRINT "Log A5= ";LA5/LOG(10)
1770 LPRINT "Log A6= ";LA6/LOG(10)
1780 LPRINT
1790 LPRINT: LPRINT
2070 LPRINT "TEMPERATURA = ";TM-273.15: LPRINT
2080 LPRINT "Log fO2=    ";LAO2
2090 LPRINT "Log fS2-py (C3)= ";LAS2
2100 LPRINT "Log fS2-py (C6)= ";LAS6
2110 LPRINT "Log fS2-po (C3)= ";LASP3
2120 LPRINT "Log fS2-po (C6)= ";LASP6
2130 LPRINT: LPRINT
2140 GOTO 30
2150 END
```

UNIVERSIDAD DE BARCELONA

Leída esta Memoria el día 9 de
enero de 1990 en la Facultad de
Geología, ante el siguiente Tribunal:

PRESIDENTE



Fig. 8.3.2.- Tall geològic del skarn de Sta. Llogaia (Sk-414). Observeu el desenvolupament del skarn en la situació secant entre el contacte intrusiu i les juntes d'estratificació.